Title stata.com

Intro 3f — Stochastic growth model

Description Remarks and examples References Also see

Description

In this introduction, we solve a stochastic growth model and interpret objects that are available after solving. The topics discussed in this introduction are more technical than those in the introductions describing other classic models. We provide some details about how dsgenl finds the model's steady state, we describe how it solves for a first-order approximation to the model's state-space form, and we provide guidance on the differences between and interpretations of linear versus log-linear approximations. For Bayesian analysis of this model, see [DSGE] Intro 9b.

Remarks and examples

stata.com

Remarks are presented under the following headings:

The model
Approximating the solution to a nonlinear DSGE model
Specifying the model to Stata
After solving
The steady state
Approximations to the policy and transition matrices

Linear and log-linear approximations

The model

The model contains equations that jointly determine output Y_t , the interest rate R_t , consumption C_t , capital K_t , and productivity Z_t . The model contains four parameters: α , β , δ , and ρ . This model is a variant on the model used in Schmitt-Grohé and Uribe (2004).

$$1 = \beta E_t \left\{ \left(\frac{C_{t+1}}{C_t} \right)^{-1} (1 + R_{t+1} - \delta) \right\}$$
 (1)

$$Y_t = Z_t K_t^{\alpha} \tag{2}$$

$$R_t = \alpha Z_t K_t^{\alpha - 1} \tag{3}$$

$$K_{t+1} = Y_t - C_t + (1 - \delta)K_t \tag{4}$$

$$\ln(Z_{t+1}) = \rho \ln(Z_t) + e_{t+1} \tag{5}$$

Equation (1) defines a relationship between consumption growth C_{t+1}/C_t and the real interest rate R_{t+1} . Equation (2) is a production function for output Y_t as a function of productivity Z_t and capital K_t . Equation (3) is a model for the interest rate. Equation (4) is the equation for capital accumulation; capital in the next period is equal to undepreciated capital this period $(1 - \delta)K_t$ plus unconsumed output $Y_t - C_t$. Equation (5) is a law of motion for productivity Z_t . The parameter β is a discount factor in the consumption equation, the parameter α is a production parameter in the output equation, the parameter δ is a depreciation parameter in the capital equation, and the parameter ρ is a persistence parameter in the productivity equation.

The state variables are the current-period capital stock and the level of productivity, (K_t, Z_t) . The control variables are consumption, the interest rate, and output (C_t, R_t, Y_t) . To solve the model means to write the control variables as functions of the state variables and to write the future values of the state variables as functions of the current state variables. Equation (2) is already solved, for example, because it writes output as a function of the state variables alone. But (4) is not a reduced form, because it writes the future capital stock as a function of the state variable K_t and the control variables C_t and Y_t .

The model equations have two key features. First, the model is nonlinear in its variables. Second, the model contains expectations of variables at time t+1, conditional on information at time t. dsgenl solves the model by taking a linear (or log-linear) approximation to the model equations at the steady state. Two concepts need to be clarified: the steady state and the approximation. We will describe these two issues in general, then return to the specific model described above.

Approximating the solution to a nonlinear DSGE model

We can gather up all the model equations into a vector of functions, f. We denote the model's vector of state variables by \mathbf{x}_t and the vector of control variables by \mathbf{y}_t . Now, the DSGE model can be written as

$$E_t \{ \mathbf{f}(\mathbf{x}_{t+1}, \mathbf{y}_{t+1}, \mathbf{x}_t, \mathbf{y}_t, \boldsymbol{\theta}) \} = \mathbf{0}$$

The expectation operator applies to the entire system of equations. In a generic DSGE model, there are n equations, n_x state variables, and n_y control variables; it must be the case that $n_x + n_y = n$.

The steady state of the model is the vector of values to which model variables converge in the absence of shocks. This is equivalent to the solution to the model's system of equations after setting all future values of variables equal to current values and dropping the expectation operator. This location is a vector of numbers $(\overline{\mathbf{x}}, \overline{\mathbf{y}})$ that solves

$$f(\overline{x}, \overline{y}, \overline{x}, \overline{y}, \theta) = 0$$

This is a system of n equations in n unknowns, and its solution is the steady state. It is this point around which we take our approximation. The location of the steady state is influenced by the model parameters θ . It is possible that the parameter vector you specify does not admit a steady state; in such a case, dsgenl will halt and issue an error message. You may specify starting values for the search for the steady state with the steadyinit() option.

We take an approximation to the model equations at the steady state. Differentiate \mathbf{f} with respect to each variable at the point $(\overline{\mathbf{x}}, \overline{\mathbf{y}}, \overline{\mathbf{x}}, \overline{\mathbf{y}})$. Differentiation generates a linearized model that is described by four matrices, namely,

$$\mathbf{F}_1 E_t \widehat{\mathbf{x}}_{t+1} + \mathbf{F}_2 E_t \widehat{\mathbf{y}}_{t+1} + \mathbf{F}_3 \widehat{\mathbf{x}}_t + \mathbf{F}_4 \widehat{\mathbf{y}}_t = \mathbf{0}$$

The matrix \mathbf{F}_1 contains partial derivatives of each model equation with respect to one-period-ahead state variables. The other \mathbf{F}_i matrices are defined similarly. Each element of each \mathbf{F}_i matrix depends on the structural parameter vector $\boldsymbol{\theta}$. There are two ways of obtaining the \mathbf{F}_i matrices: linear approximation and log-linear, or percentage, approximation. The "hat" notation denotes the variables of the linear model. In a linear approximation,

$$\hat{\mathbf{x}}_t = (\mathbf{x}_t - \overline{\mathbf{x}})$$

that is, $\hat{\mathbf{x}}_t$ is the unit difference from steady state. In a log-linear approximation,

$$\widehat{\mathbf{x}}_t = (\mathbf{x}_t - \overline{\mathbf{x}})/\overline{\mathbf{x}}$$

so that $\hat{\mathbf{x}}_t$ is measured in percentage difference from the steady state. Which approximation is applied will affect the \mathbf{F}_i matrices, parameter estimation, and the interpretation of all objects available after estimation.

Hatted variables are measured as deviations from steady state. At this point, the model is in the form of a linear DSGE model, and the methods of Klein (2000) may be employed to find the solution. The stability conditions of Klein (2000) apply to the \mathbf{F}_i matrices, and dsgen1 will not proceed if the stability conditions are not met. If the stability conditions are met, we exactly solve the linear model. The solution is of the form

$$\hat{\mathbf{y}}_t = \mathbf{G}\hat{\mathbf{x}}_t$$

$$\hat{\mathbf{x}}_{t+1} = \mathbf{H}\hat{\mathbf{x}}_t + \mathbf{M}\mathbf{e}_{t+1}$$

Because shocks \mathbf{e}_t are mean zero, it does not matter whether we write \mathbf{e}_t or $\hat{\mathbf{e}}_t$.

In a nonlinear DSGE model, the linear solution matrices are a linear approximation to the true nonlinear solution functions.

$$\mathbf{y}_t = \mathbf{g}(\mathbf{x}_t)$$

 $\mathbf{x}_{t+1} = \mathbf{h}(\mathbf{x}_t) + \mathbf{M}\mathbf{e}_{t+1}$

The matrix G is an approximation to the function $g(\cdot)$, and the matrix H is an approximation to the function $h(\cdot)$. Once the solution has been obtained, we can view the steady state, the approximate solution matrices, the impulse responses, and other objects.

Specifying the model to Stata

We import U.S. macro data. Although we are solving the model (hence, the data will play no role), dsgen1 expects there to be data anyway.

```
. use https://www.stata-press.com/data/r18/usmacro2
(Federal Reserve Economic Data - St. Louis Fed, 2017-01-15)
```

Next, we set up a parameter matrix containing the values that we will use for our model. We type

- $. matrix param_mat = (0.33, 0.96, 0.025, 0.9, 1)$
- . matrix colnames param_mat = alpha beta delta rho /sd(e.z)

These values are similar to those in King and Rebelo (1999) and Schmitt-Grohé and Uribe (2004). The elasticity of output with respect to capital α is set to 0.33. The discount factor β is set to 0.96. The depreciation rate δ is set to 0.025. The persistence of productivity is set to 0.9, and the standard deviation of the shock to productivity is set to 1.

Next, we specify the model.

У	Coefficient	Std. err.	z	P> z	[95% conf.	interval]
/structural						
beta	.96				•	
delta	.025				•	
alpha	.33				•	
rho	.9	•	•			
sd(e.z)	1	•			•	

Number of obs = 244

Note: Skipped identification check.

Note: Model solved at specified parameters; maximization options ignored.

Each equation is bound in parentheses. Variables appear without any special modification; parameters are bound in parentheses. After the equation list, we specify seven options. The first four tell dsgen1 how to interpret each variable in the model. The control variables y, r, and c are specified as being either observed() or unobserved(). Because we are only solving the model, it is inconsequential how we place control variables into these categories. In an estimation exercise, data on observed control variables are used to estimate model parameters, and which variables are observable can affect parameter identification. The state variables are divided into categories according to whether they are subject to shocks. The state variable k is not subject to shocks; it is an endogenous state variable. The state variable z is subject to shocks; it is an exogenous state variable.

The from() option specifies the starting values for the parameters. The solve option declares that we wish to solve the model at the values in from(). The noidencheck option skips the check for local identification.

The output table repeats the parameter vector that we passed in; it provides no new information. The notes below the estimation table remind us that we have skipped the identification check and that we have solved the model rather than estimated the model parameters.

After solving

After solving the model by running dsgenl, you can view the steady state, the approximate policy and transition matrices, and more using the estat commands.

The steady state

In the process of finding the solution, dsgen1 finds the steady state of the model. The steady state is displayed with

. estat steady, compact

Location of model steady-state

	Coefficient
k	10.88
z	1.00
С	1.93
r	0.07
у	2.20

Because we used dsgenl's solve option, estat steady displays the steady state of the model under the initial parameter vector. When solve is not specified, estat steady displays the steady state of the model at the converged parameter values.

The steady-state level of the capital stock is 10.88 units, and the steady-state level of output is 2.20 units. The steady-state interest rate is 0.07, or 7%. Some of these values are more meaningful when expressed as ratios rather than when they are expressed as raw values. The model's consumption-to-output ratio is about 0.88. The model's capital-to-output ratio is about 4.9.

Approximations to the policy and transition matrices

Next, we approximate the policy and state transition matrices. estat policy provides the approximate policy matrix.

. estat policy, compact

Policy matrix

	k	z
c r	.6077069 67	.3431366
У	.33	1

By default, dsgenl uses a log-linear approximation, and the results of estat policy are also based on this approximation. This means that variables are measured as percentage deviations from steady state. The interpretation of the output equation in the policy matrix is that

$$\frac{Y_t - \overline{Y}}{\overline{Y}} \approx 0.33 \frac{K_t - \overline{K}}{\overline{K}} + \frac{Z_t - \overline{Z}}{\overline{Z}}$$

A log-linear approximation was chosen because many DSGE model equations are linear in logs, or nearly so. For instance, in logs, (2) is

$$\log(Y_t) = \alpha \log(K_t) + \log(Z_t)$$

just as the policy matrix describes.

The entries in the policy matrix are impact effects. They display the percentage change in a control variable that results from a one-percent change in a state variable. From the consumption equation, a 1% increase in z causes consumption to rise by 34%. From the output equation, a 1% increase in z causes output to rise by 1%.

estat transition provides the approximate state transition matrix.

. estat transition, compact

Transition matrix of state variables

	k	z
k F1.	.9340903	.1412781
z F1.	0	.9

The state transition matrix displays how state variables in the next period depend on state variables today. A 1% increase in z causes capital to rise by 0.14% one period in the future. Notice that an increase in z has an effect on k, but an increase in k has no cross-effect on z.

Linear and log-linear approximations

Specifying option linearapprox causes dsgenl to take a linear approximation rather than a log-linear approximation. We first re-solve the model.

```
. dsgenl (1 = {beta}*(F.c/c)^(-1)*(1+F.r-{delta})) 

> (r = {alpha}*z*k^({alpha}-1)) 

> (y = z*k^{alpha}) 

> (F.k = y - c + (1-{delta})*k) 

> (ln(F.z) = {rho}*ln(z)) 

> , observed(y) unobserved(c r) exostate(z) endostate(k) 

> from(param_mat) solve noidencheck linearapprox 

Solving at initial parameter vector ...
```

First-order DSGE model

Sample: 1955q1 thru 2015q4 Log likelihood = -808.47167 Number of obs = 244

у	Coefficient	Std. err.	z	P> z	[95% conf.	interval]
/structural	00					
beta delta	.96 .025	•	•	•	•	•
alpha	.025	•	•	•	•	•
rho	.9	•	•	•	•	
				<u>-</u>		
sd(e.z)	1	•			•	

Note: Skipped identification check.

Note: Model solved at specified parameters; maximization options ignored.

We next examine the impact that the linearapprox option has on postsolution objects. The choice of linear or log-linear approximation does not affect the location of the steady state.

. estat steady, compact

Location of model steady-state

	Coefficient
k z c r	10.88 1.00 1.93 0.07 2.20

This is to be expected; the steady state is the point around which approximations are taken, but its location does not depend on the choice of approximation.

The policy matrix has changed.

. estat policy, compact

Policy matrix

	k	z
c	.1075764	.6610193
r	0041045	.0666667
y	.0666667	2.198462

Each entry displays the unit increase in a control variable after a unit increase in a state variable. As an example, we use the estimates in the last line of this table. The interpretation of the output equation is

$$(Y_t - \overline{Y}) \approx 0.067(K_t - \overline{K}) + 2.2(Z_t - \overline{Z})$$

so that a one-unit increase in the capital stock leads output to rise by 0.067 units. The other entries are interpreted analogously.

The state transition matrix is now

. estat transition, compact

Transition matrix of state variables

	k	z
k F1.	.9340903	1.537443
z F1.	0	.9

The approximate capital accumulation equation is

$$(K_{t+1} - \overline{K}) \approx 0.93(K_t - \overline{K}) + 1.54(Z_t - \overline{Z})$$

so that a one-unit increase in productivity Z_t leads to a 1.54-unit increase in the capital stock in the next period.

o intro

References

King, R. G., and S. T. Rebelo. 1999. Resuscitating real business cycles. In *Handbook of Macroeconomics: Volume 1A*, ed. J. B. Taylor and M. Woodford, 927–1007. New York: Elsevier. https://doi.org/10.1016/S1574-0048(99)10022-3.

Klein, P. 2000. Using the generalized Schur form to solve a multivariate linear rational expectations model. *Journal of Economic Dynamics and Control* 24: 1405–1423. https://doi.org/10.1016/S0165-1889(99)00045-7.

Schmitt-Grohé, S., and M. Uribe. 2004. Solving dynamic general equilibrium models using a second-order approximation to the policy function. *Journal of Economic Dynamics and Control* 28: 755–775. https://doi.org/10.1016/S0165-1889(03)00043-5.

Also see

[DSGE] **Intro** 1 — Introduction to DSGEs

[DSGE] Intro 3d — Nonlinear New Keynesian model

[DSGE] Intro 3e — Nonlinear New Classical model

[DSGE] Intro 9b — Bayesian estimation of stochastic growth model

[DSGE] **dsgenl** — Nonlinear dynamic stochastic general equilibrium models

[DSGE] dsgenl postestimation — Postestimation tools for dsgenl

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.